

VBA（注意保存为宏）

批量创建工作表

在excel表格中输入标题，注意要字符形式，数字和日期形式可以在最前面加单引号。visual basic-sheet1右键-插入-模块-输入代码

	A	B
1	标题	
2	C6675	
3	C6676	
4	C6677	
5	C6678	
6	C6679	
7	C6680	
8	C6681	
9	C6682	
10	C6683	
11	C6684	
12	C6685	
13	C6686	

代码：

代码块

```
1 Sub NewSht()  
2     Dim shtActive As Worksheet, sht As Worksheet  
3     Dim i As Long, strShtName As String  
4     On Error Resume Next '当代码出错时继续运行  
5     Set shtActive = ActiveSheet  
6     For i = 2 To shtActive.Cells(Rows.Count, 1).End(xlUp).Row  
7         '单元格A1是标题，跳过，从第2行开始遍历工作表名称  
8         strShtName = shtActive.Cells(i, 1).Value  
9         '工作表名强制转换为字符串类型  
10        Set sht = Sheets(strShtName)  
11        '当工作簿不存在工作表Sheets(strShtName)时，这句代码会出错，然后.....  
12        If Err Then  
13            '如果代码出错，说明不存在工作表Sheets(t)，则新建工作表  
14            Worksheets.Add , Sheets(Sheets.Count)  
15            '新建一个工作表，位置放在所有已存在工作表的后面  
16            ActiveSheet.Name = strShtName  
17            '新建的工作表必然是活动工作表，为之命名  
18            Err.Clear  
19            '清除错误状态
```

```
20         End If
21     Next
22     shtActive.Activate
23     '重新激活原工作表
24 End Sub
```

删除工作表只留最后一张

代码：

代码块

```
1 Sub DelShet() '删除所有工作表
2     Dim sht As Worksheet
3     Application.ScreenUpdating = False '关屏幕刷新
4     Application.DisplayAlerts = False '关警告信息
5     On Error Resume Next
6     For Each sht In Worksheets
7         sht.Delete '遍历工作表删除
8     Next
9     Application.ScreenUpdating = True
10    Application.DisplayAlerts = True
11 End Sub
```

提取所有工作表名字

代码：

代码块

```
1 Sub GetShtByVba()
2     Dim sht As Worksheet, k As Long
3     Application.ScreenUpdating = False
4     k = 1
5     Range("a:b").Clear '清空数据
6     Range("a:a").NumberFormat = "@" '设置文本格式
7     For Each sht In Worksheets '遍历工作表取表名
8         k = k + 1
9         Cells(k, 1) = sht.Name
10    Next
11    Range("a1:b1") = Array("工作表名", "是否删除")
```

```
12 Application.ScreenUpdating = True
13 End Sub
```

删除指定名字的工作表

要先提取所有工作表的名字，然后在是否删除那一列写“删除”，然后新建模块输入代码

代码块

```
1 Sub DelShtByVba()
2     Dim sht As Worksheet, i As Long, r
3     Application.ScreenUpdating = False
4     Application.DisplayAlerts = False
5     On Error Resume Next
6     r = Range("a1").CurrentRegion '数据装入数组r
7     For i = 2 To UBound(r) '遍历并删除工作表
8         If r(i, 2) = "删除" Then Worksheets(CStr(r(i, 1))).Delete
9     Next
10    Application.ScreenUpdating = True
11    Application.DisplayAlerts = True
12 End Sub
```

生成带超链接的工作表目录

代码块

```
1 Dim strShtName As String
2 Sub Mybutton()
3     Dim sht As Worksheet, btn As Button
4     On Error Resume Next
5     For Each sht In Worksheets
6         With sht
7             If .Name <> strShtName Then
8                 .Shapes(strShtName).Delete
9                 '删除原有的名称为shtn的按钮，避免重复创建
10                Set btn = .Buttons.Add(0, 0, 60, 30) '使用add方法在工作表中添加一个按钮控件，add方法语法如下:表达式.Add(left,right,width,height)
11                '新建按钮，释义见小贴士
12                With btn
13                    .Name = strShtName
```

```

14          '命令按钮命名
15          .Characters.Text = "返回总表"
16          '按钮的文本内容
17          .OnAction = "LinkTable"
18          '指定按钮控件所执行的宏命令
19      End With
20  End If
21 End With
22 Next
23 Set btn = Nothing
24 End Sub
25
26 Sub LinkTable()
27     strShtName = "总表" '指定了返回总表的名字，可以根据实际需要修改为目标表的名称，比
    如“目录”。
28     '设置变量strShtName为总表的名称，可以根据实际总表的名称做修改
29     Worksheets(strShtName).Activate
30     [a1].Select
31 End Sub
32

```

在各分表创建返回总表的超链接

要注意代码中的“总表”

代码块

```

1  Dim strShtName As String
2  Sub Mybutton()
3      Dim sht As Worksheet, btn As Button
4      On Error Resume Next
5      For Each sht In Worksheets
6          With sht
7              If .Name <> strShtName Then
8                  .Shapes(strShtName).Delete
9                  '删除原有的名称为shtn的按钮，避免重复创建
10                 Set btn = .Buttons.Add(0, 0, 60, 30) '使用add方法在工作表中添加一
    个按钮控件，add方法语法如下:表达式.Add(left,right,width,height)
11                 '新建按钮，释义见小贴士

```

```

12         With btn
13             .Name = strShtName
14             '命令按钮命名
15             .Characters.Text = "返回总表"
16             '按钮的文本内容
17             .OnAction = "LinkTable"
18             '指定按钮控件所执行的宏命令
19         End With
20     End If
21 End With
22 Next
23 Set btn = Nothing
24 End Sub
25
26 Sub LinkTable()
27     strShtName = "总表" '指定了返回总表的名字，可以根据实际需要修改为目标表的名称，比
    如“目录”。
28     '设置变量strShtName为总表的名称，可以根据实际总表的名称做修改
29     Worksheets(strShtName).Activate
30     [a1].Select
31 End Sub
32

```

批量提取工作表名字

不会出现是否删除那一列

代码块

```

1 Sub GetShtName()
2     Dim sht As Worksheet, i As Long
3     i = 1 'i初始值为1
4     With Columns(1)
5         .ClearContents '清除A列内容
6         .NumberFormat = "@" '设置单元格格式为文本
7     End With
8     Cells(1, 1) = "工作表名称目录"
9     For Each sht In Worksheets '遍历工作表
10         i = i + 1
11         Cells(i, 1) = sht.Name '在A列记录工作表名称
12     Next
13 End Sub

```

批量修改工作表名字

在上述提取之后再修改，之间在提取出来的工作表列表第二列输入对应的名字

代码块

```
1 Sub ReNameSht()  
2     Dim strShtName$, sht As Worksheet, i&  
3     On Error Resume Next '当程序运行中出现错误时，继续运行  
4     For i = 2 To Cells(Rows.Count, 1).End(xlup).Row '遍历当前表格A列的数据  
5         strShtName = Cells(i, 1).Value '将表格A列的值，赋予变量strShtName  
6         Worksheets(strShtName).Name = Cells(i, 2).Value '工作表重命名  
7     Next  
8 End Sub
```

批量取消工作簿隐藏

代码块

```
1 Sub unShtVisible()  
2     Dim sht As Worksheet  
3     For Each sht In Worksheets '遍历工作表，设置可见  
4         sht.Visible = xlSheetVisible  
5     Next  
6 End Sub
```

汇总分表成总表（不保留分表格式）

代码块

```
1 Sub CollectData_Fixed()  
2     Dim Sht As Worksheet, rng As Range, k&, n&  
3     Dim targetRow As Long  
4  
5     Application.ScreenUpdating = False  
6  
7     n = Val(InputBox("请输入标题的行数", "提醒"))  
8     If n < 0 Then MsgBox "标题行数不能为负数。", 64, "提示": Exit Sub  
9  
10    Cells.ClearContents  
11    k = 0  
12  
13    For Each Sht In Worksheets  
14        If Sht.Name <> ActiveSheet.Name Then  
15            With Sht  
16                lastRow = .Cells(.Rows.Count, 1).End(xlUp).Row
```

```

17         lastCol = .Cells(1, .Columns.Count).End(xlToLeft).Column
18
19     If lastRow > 0 And lastCol > 0 Then
20         Set rng = .Range(.Cells(1, 1), .Cells(lastRow, lastCol))
21
22         k = k + 1
23
24         If k = 1 Then
25             '第一个表：复制全部（包括标题）
26             rng.Copy
27             [a1].PasteSpecial Paste:=xlPasteValues
28         Else
29             '其他表：跳过标题行，只复制数据
30             If lastRow > n Then
31                 Set rng = .Range(.Cells(n + 1, 1), .Cells(lastRow,
lastCol))
32
33                 '精确定位目标位置
34                 With ActiveSheet
35                     targetRow = .Cells(.Rows.Count,
1).End(xlUp).Row + 1
36                     If targetRow = 2 And .Range("A2") = "" Then
targetRow = 1
37                     End With
38
39                     rng.Copy
40                     Cells(targetRow, 1).PasteSpecial
Paste:=xlPasteValues
41                     End If
42                 End If
43             End If
44         End With
45     End If
46 Next
47
48 [a1].Activate
49 Application.CutCopyMode = False
50 Application.ScreenUpdating = True
51 End Sub

```

汇总分表成总表（保留分表格式）

```

1 Sub CollectDataFromShtFormat()
2     Dim sht As Worksheet, rng As Range, k As Long, nTitleCount As Long
3     On Error Resume Next
4     nTitleCount = Val(InputBox("请输入标题的行数", "提醒", 1))
5     If nTitleCount < 0 Then MsgBox "标题行数不能为负数。", 64, "提示": Exit Sub
6     Application.ScreenUpdating = False
7     Cells.ClearContents '清空当前表数据
8     For Each sht In Worksheets '遍历工作表
9         If sht.Name <> ActiveSheet.Name Then
10             '如果工作表名称不等于当前表名则进行汇总动作.....
11             Set rng = sht.UsedRange
12             k = k + 1 '累计K值
13             If k = 1 Then '如果是首个表格，则K为1，则把标题行一起复制到汇总表
14                 sht.Cells.Copy: Range("a1").PasteSpecial Paste:=xlPasteFormats
15                 '只粘贴格式
16                 rng.Copy: Range("a1").PasteSpecial Paste:=xlPasteValues '只粘贴
17                 数值
18             Else '否则，扣除标题行后再复制黏贴到总表，只黏贴数值
19                 rng.Offset(nTitleCount).Copy
20                 With Cells(ActiveSheet.UsedRange.Rows.Count + 1, 1)
21                     .PasteSpecial Paste:=xlPasteFormats '粘贴格式
22                     .PasteSpecial Paste:=xlPasteValues '粘贴数值
23                 End With
24             End If
25         End If
26     Next
27     Range("a1").Activate
28     Application.ScreenUpdating = True '恢复屏幕刷新
29     MsgBox "汇总OK，一共汇总了：" & k & "张工作表"
30 End Sub

```

对工作表进行批量排序

先提取名字

代码块

```

1 Sub GetShtName()
2     Dim k As Long, sht As Worksheet
3     Application.ScreenUpdating = False
4     With Columns(1)
5         .ClearContents '清空A列原有数据
6         .NumberFormat = "@" '设置单元格格式为文本
7     End With
8     Cells(1, 1) = "目录"
9     k = 1

```



```

10 For Each sht In ThisWorkbook.Worksheets '遍历工作表
11     If sht.Name <> ActiveSheet.Name Then '如果sht不等于当前工作表名称
12         k = k + 1 '累加工作表个数
13         Cells(k, 1) = sht.Name '工作表名称写入A列
14     End If
15 Next
16 Application.ScreenUpdating = True
17 End Sub

```

代码块

```

1 Sub SortSht()
2     Dim shtActive As Worksheet, i As Long
3     Dim arr, strShtName As String
4     On Error Resume Next
5     Application.ScreenUpdating = False
6     Set shtActive = ActiveSheet '当前表赋值变量shtactive
7     arr = Range("a1:a" & Cells(Rows.Count, 1).End(xlUp).Row)
8     'A列数据装入数组arr
9     For i = 2 To UBound(arr) '遍历数组arr
10         strShtName = arr(i, 1)
11         Worksheets(strShtName).Move after:=Worksheets(i - 1)
12         '指定工作表按顺序排放
13     Next
14     shtActive.Select '回到操作表
15     Application.ScreenUpdating = True
16 End Sub

```

批量工作表加密(可以进去但只让看不让改)

代码块

```

1 Sub ProtectSht()
2     Dim strAds As String, sht As Worksheet
3     Dim strKey As String, strTemp As String
4     Dim rng As Range, strMsg As String
5     Dim strNoShtName As String, strYesShtName As String
6     On Error Resume Next
7     strAds = InputBox("请输入单元格保存范围, 例如A1:B10." & vbCr _
8         & "可以设置不连续单元格, 中间请以逗号分隔。比如
9         A1:B10,D2:D8" & vbCr _
10        & "如果需要全表保护, 可以直接确定。", Default:="全表
11        保护")
12     If StrPtr(strAds) = False Then Exit Sub
13     If strAds = "全表保护" Then strAds = Cells.Address

```

```

12 Set rng = Range(strAds) '测试输入的单元格区域是否有效
13 If Err Then MsgBox "你输入的单元格区域地址不是正确的格式，请重新操作。": Exit Sub
14 strKey = InputBox("请输入保护密码。") '第一次输入密码
15 If StrPtr(strKey) = False Then Exit Sub
16 strTemp = InputBox("请再次输入保护密码。") '第二次输入密码
17 If StrPtr(strKey) = False Then Exit Sub
18 If strKey <> strTemp Then MsgBox "你两次输入的密码不一致，系统退出，请重新操作。": Exit Sub
19 For Each sht In Worksheets '遍历工作表加密保护
20     With sht
21         If .ProtectContents = False Then '如果工作表未保护
22             .Cells.Locked = False '全部单元格区域取消锁定
23             .Range(strAds).Locked = True '需要保护的区域锁定
24             .Protect strKey, True, True, True '保护工作表，只允许编辑非锁定区域
25             strYesShtName = strYesShtName & "," & .Name '保护成功的工作表名称
26         Else
27             strNoShtName = strNoShtName & "," & .Name '自身已有保护功能的工作表
28         End If
29     End With
30 Next
31 If strYesShtName <> "" Then strMsg = "工作表：" & Mid(strYesShtName, 2) & "的" & strAds & "区域保护完成"
32 If strNoShtName <> "" Then strMsg = strMsg & vbCrLf & "以下工作表自身已有保护，无法再次保护：" & Mid(strNoShtName, 2)
33 MsgBox (strMsg)
34 End Sub

```

批量工作表解密

代码块

```

1 Sub UnProtct()
2     MsgBox "破解提示：当要求输入密码时请点击取消！"
3     Application.DisplayAlerts = False
4     On Error Resume Next
5     Dim sht As Worksheet
6     For Each sht In Worksheets
7         With sht
8             .Protect DrawingObjects:=True, Contents:=True, Scenarios:=True,
AllowFiltering:=True, AllowUsingPivotTables:=True
9             .Protect DrawingObjects:=False, Contents:=True, Scenarios:=False,
AllowFiltering:=True, AllowUsingPivotTables:=True
10            .Protect DrawingObjects:=True, Contents:=True, Scenarios:=False,
AllowFiltering:=True, AllowUsingPivotTables:=True

```

```

11         .Protect DrawingObjects:=False, Contents:=True, Scenarios:=True,
        AllowFiltering:=True, AllowUsingPivotTables:=True
12         .Unprotect
13     End With
14 Next
15     MsgBox "ok"
16 End Sub

```

按任意列拆分工作表

就比如有n条明细数据，但是各部门的都杂糅在里面，我想按照每个部门建一个表，可以采用这个方式

A	B	C	
年份	日期	部门	
1999	1月	财务部	
2000	2月	市场部	
2001	3月	财务部	
2002	4月	采购部	
2003	5月	财务部	
2004	6月	财务部	
2005	7月	市场部	
2006	8月	财务部	
2007	9月	采购部	
2008	10月	财务部	
2009	11月	财务部	
2010	12月	市场部	
2011	13月	财务部	

代码块

```

1 Sub SplitShts()
2     Dim d As Object, sht As Worksheet
3     Dim aData, aResult, aTemp, aKeys, i&, j&, k&, x&
4     Dim rngData As Range, rngGist As Range
5     Dim lngTitleCount&, lngGistCol&, lngColCount&
6     Dim rngFormat As Range, aRef, strYesOrNo As String
7     Dim strKey As String, strTemp As String
8
9     On Error Resume Next
10    Set d = CreateObject("scripting.dictionary")
11    Set rngGist = Application.InputBox("请框选拆分依据列！ 只能选择单列单元格区域！", Title:="提示", Type:=8)
12    lngGistCol = rngGist.Column

```

```
13     lngTitleCount = Val(Application.InputBox("请输入总表标题行的行数? ",
14     Default:=1))
15
16     If lngTitleCount < 0 Then
17         MsgBox "标题行数不能为负数，程序退出。":
18         Exit Sub
19     End If
20
21     strYesOrNo = MsgBox("是否需要在分表保留总表格式? ", vbYesNo)
22     Set rngData = rngGist.Parent.UsedRange
23     Set rngFormat = rngGist.Parent.Cells
24     aData = rngData.Value
25     lngGistCol = lngGistCol - rngData.Column + 1
26     lngColCount = UBound(aData, 2)
27
28     Application.ScreenUpdating = False
29     Application.DisplayAlerts = False
30
31     ReDim aRef(1 To UBound(aData))
32     For i = 1 To UBound(aData)
33         If IsError(aData(i, lngGistCol)) Then
34             aRef(i) = "错误值"
35         ElseIf aData(i, lngGistCol) = "" Then
36             strTemp = ""
37             For j = 1 To lngColCount
38                 strTemp = strTemp & aData(i, j)
39             Next
40             If strTemp = "" Then
41                 aRef(i) = "整行空白"
42             Else
43                 aRef(i) = "空白单元格"
44             End If
45         Else
46             strKey = aData(i, lngGistCol)
47             aRef(i) = strKey
48         End If
49     Next
50
51     For i = lngTitleCount + 1 To UBound(aData)
52         strKey = aRef(i)
53         If strKey <> "整行空白" Then
54             If Not d.exists(strKey) Then
55                 d(strKey) = ""
56                 ReDim aResult(1 To UBound(aData), 1 To lngColCount)
57                 k = 0
58                 '复制标题行数据
```

```

59         For x = 1 To lngTitleCount
60             k = k + 1
61             For j = 1 To lngColCount
62                 aResult(k, j) = aData(x, j)
63             Next
64         Next
65
66         '复制数据行
67         For x = lngTitleCount + 1 To UBound(aData)
68             strTemp = aRef(x)
69             If strTemp = strKey Then
70                 k = k + 1
71                 For j = 1 To lngColCount
72                     aResult(k, j) = aData(x, j)
73                 Next
74             End If
75         Next
76
77         For Each sht In ActiveWorkbook.Worksheets
78             If sht.Name = strKey Then sht.Delete
79         Next
80
81         With Worksheets.Add(, Sheets(Sheets.Count))
82             .Name = strKey
83             .Range("a1").Resize(UBound(aData),
120         lngColCount).NumberFormat = "@"
84
85             '写入标题和数据（修正的部分）
86             .Range("a1").Resize(k, lngColCount) = aResult
87
88             If strYesOrNo = vbYes Then
89                 rngFormat.Copy
90                 .Range("a1").PasteSpecial Paste:=xlPasteFormats,
120         Operation:=xlNone, SkipBlanks:=False, Transpose:=False
91                 .Range("a1").Offset(k, 0).Resize(UBound(aData) - k,
120         1).EntireRow.Delete
92             End If
93             .Range("a1").Select
94         End With
95     End If
96 End If
97 Next
98
99     rngData.Parent.Activate
100     Application.ScreenUpdating = True
101     Application.DisplayAlerts = True
102     Set d = Nothing

```

```

103     Set rngData = Nothing
104     Set rngGist = Nothing
105     Set rngFormat = Nothing
106     Erase aData: Erase aResult
107     MsgBox "数据拆分完成! "
108 End Sub

```

批量将工作表转成独立的工作簿

代码块

```

1 Sub EachShtToWorkbook()
2     Dim sht As Worksheet, strPath As String
3     With Application.FileDialog(msoFileDialogFolderPicker)
4         '选择保存工作簿的文件路径
5         If .Show Then strPath = .SelectedItems(1) Else Exit Sub
6         '读取选择的文件路径,如果用户未选取路径则退出程序
7     End With
8     If Right(strPath, 1) <> "\" Then strPath = strPath & "\"
9     Application.DisplayAlerts = False
10    '取消显示系统警告和消息,避免重名工作簿无法保存。当有重名工作簿时,会直接覆盖保存。
11    Application.ScreenUpdating = False '取消屏幕刷新
12    For Each sht In Worksheets '遍历工作表
13        sht.Copy '复制工作表,工作表单纯复制后,会成为活动工作簿
14        With ActiveWorkbook
15            .SaveAs strPath & sht.Name, xlWorkbookDefault
16            '保存活动工作簿到指定路径下,以当前系统默认文件格式
17            .Close True '关闭工作簿并保存
18        End With
19    Next
20    MsgBox "处理完成。", , "提醒"
21    Application.ScreenUpdating = True '恢复屏幕刷新
22    Application.DisplayAlerts = True '恢复显示系统警告和消息
23 End Sub

```

将总表按任意列拆分成多个工作簿

代码块

```

1 Sub SplitShts()
2     Dim d As Object, sht As Worksheet
3     Dim aData, aResult, aTemp, aKeys, i&, j&, k&, x&

```

```

4 Dim rngData As Range, rngGist As Range, ws As Workbook
5 Dim lngTitleCount&, lngGistCol&, lngColCount&
6 Dim rngFormat As Range, aRef, strYesOrNo As String
7 Dim strKey As String, strTemp As String, strPath As String
8 On Error Resume Next '忽略错误, 程序继续运行
9 Set d = CreateObject("scripting.dictionary")
10 With Application.FileDialog(msoFileDialogFolderPicker)
11 '用户选择保存工作簿的路径
12 If .Show Then strPath = .SelectedItems(1) Else Exit Sub
13 End With
14 If Right(strPath, 1) <> "\" Then strPath = strPath & "\"
15 Set rngGist = Application.InputBox("请框选拆分依据列! 只能选择单列单元格区
域!", Title:="提示", Type:=8)
16 '用户选择的拆分依据列
17 If rngGist Is Nothing Then Exit Sub
18 lngGistCol = rngGist.Column '拆分依据列的列标
19 lngTitleCount = Val(Application.InputBox("请输入总表标题行的行数? ",
Default:=1))
20 '用户设置总表的标题行数
21 If lngTitleCount < 0 Then MsgBox "标题行数不能为负数, 程序退出。": Exit Sub
22 strYesOrNo = MsgBox("是否需要在分表保留总表格式?", vbYesNo)
23 Set rngData = rngGist.Parent.UsedRange
24 '总表的数据区域
25 Set rngFormat = rngGist.Parent.Cells
26 '总表的单元格区域用于粘贴总表格式
27 aData = rngData.Value '数据源装入数组
28 lngGistCol = lngGistCol - rngData.Column + 1
29 '计算依据列在数组中的位置
30 lngColCount = UBound(aData, 2)
31 '数据源的列数
32 Application.ScreenUpdating = False
33 Application.DisplayAlerts = False
34 ReDim aRef(1 To UBound(aData))
35 For i = 1 To UBound(aData) '处理依据列的异常值, 空白/错误值/整行空白等
36 If IsError(aData(i, lngGistCol)) Then
37 aRef(i) = "错误值"
38 ElseIf aData(i, lngGistCol) = "" Then
39 strTemp = "" '判断是否整行数据为空
40 For j = 1 To lngColCount
41 strTemp = strTemp & aData(i, j)
42 Next
43 If strTemp = "" Then '如果整行为空
44 aRef(i) = "整行空白"
45 Else
46 aRef(i) = "空白单元格"
47 End If
48 Else

```

```

49         strKey = aData(i, lngGistCol)
50         aRef(i) = strKey
51     End If
52 Next
53 For i = lngTitleCount + 1 To UBound(aData)
54     strKey = aRef(i)
55     If strKey <> "整行空白" Then
56         If Not d.exists(strKey) Then
57             '字典中不存在关键字时则遍历建表
58             d(strKey) = ""
59             ReDim aResult(1 To UBound(aData), 1 To lngColCount) '声明一个结果
数组
60             k = 0
61             For x = lngTitleCount + 1 To UBound(aData) '遍历数据源
62                 strTemp = aRef(x)
63                 If strTemp = strKey Then '如果记录符合条件，则装入结果数组
64                     k = k + 1
65                     For j = 1 To lngColCount
66                         aResult(k, j) = aData(x, j)
67                     Next
68                 End If
69             Next
70             Set ws = Workbooks.Add
71             With ws.Sheets(1)
72                 '新建一个工作簿
73                 .Range("a1").Resize(UBound(aData),
lngColCount).NumberFormat = "@"
74                 '设置单元格为文本格式
75                 If lngTitleCount > 0 Then
76                     .Range("a1").Resize(lngTitleCount, lngColCount) = aData
77                     '标题行
78                     .Range("a1").Offset(lngTitleCount, 0).Resize(k,
lngColCount) = aResult
79                     '写入数据
80                     If strYesOrNo = vbYes Then '如果用户选择保留总表格式
81                         rngFormat.Copy
82                         .Range("a1").PasteSpecial Paste:=xlPasteFormats,
Operation:=xlNone, SkipBlanks:=False, Transpose:=False
83                         '复制粘贴总表的格式
84                         .Range("a1").Offset(lngTitleCount + k,
0).Resize(UBound(aData) - k - lngTitleCount, 1).EntireRow.Delete
85                         '删除多余的格式单元格
86                     End If
87                     .Range("a1").Select
88                 End With
89                 ws.SaveAs strPath & strKey, xlWorkbookDefault
ws.Close False

```



```

90         End If
91     End If
92 Next
93 Application.ScreenUpdating = True
94 Application.DisplayAlerts = True
95 Set d = Nothing
96 Set rngData = Nothing
97 Set rngGist = Nothing
98 Set rngFormat = Nothing
99 Erase aData: Erase aResult
100 MsgBox "数据拆分完成！"
101 End Sub

```

选中行或列自动填充颜色

双击thisworkbook直接输代码，不用运行，直接叉掉

代码块

```

1 Private Sub Workbook_SheetSelectionChange(ByVal Sh As Object, ByVal Target As Range)
2     Application.ScreenUpdating = False
3     Cells.Interior.ColorIndex = -4142 '取消单元格原有填充色，但不包含条件格式产生的颜色。
4     Rows(Target.Row).Interior.ColorIndex = 33 '活动单元格整行填充颜色
5     Columns(Target.Column).Interior.ColorIndex = 33 '活动单元格整列填充颜色
6     Application.ScreenUpdating = True
7 End Sub

```

将指定名称批量创建工作簿

需要将工作簿得名称在总表中输入，第一行是标题，不会被拆分

代码块

```

1 Sub CreateFiles()
2     Dim strPath As String, strFileName As String
3     Dim i As Long, r
4     On Error Resume Next
5     With Application.FileDialog(msoFileDialogFolderPicker)
6         '用户选择文件夹路径
7         If .Show Then strPath = .SelectedItems(1) Else Exit Sub
8         '如果用户为选择文件夹则退出程序

```

```

9      End With
10     If Right(strPath, 1) <> "\" Then strPath = strPath & "\"
11     Application.ScreenUpdating = False '取消屏幕刷新
12     Application.DisplayAlerts = False '取消警告提示, 当有重名工作簿时直接覆盖
13     r = Range("a1:a" & Cells(Rows.Count, 1).End(xlUp).Row) '数据装入数组r
14     For i = 2 To UBound(r) '标题不要, 因此从第2个元素开始遍历数组r
15         With Workbooks.Add '新建工作簿
16             .SaveAs strPath & r(i, 1), xlWorkbookDefault
17             '以指定名称、默认文件类型保存工作簿
18             .Close True '关闭工作簿
19         End With
20     Next
21     Application.ScreenUpdating = True
22     Application.DisplayAlerts = True
23     MsgBox "创建完成。"
24 End Sub

```

在一个文件夹下按指定条件批量删除工作簿

新建一个空表, 输入下列代码, 然后就会在这个新表里面得出哪个文件夹下所有工作簿得名称

代码块

```

1  Sub GetFiles()
2      Dim strPath As String, strFileName As String, k As Long
3      With Application.FileDialog(msoFileDialogFolderPicker)
4          If .Show Then strPath = .SelectedItems(1) Else: Exit Sub
5          '获取用户选择的文件夹的路径, 如果未选取, 则退出程序
6      End With
7      If Right(strPath, 1) <> "\" Then strPath = strPath & "\"
8      Application.ScreenUpdating = False
9      Range("a:b").Clear: k = 1
10     '清除A:B列的所有
11     Cells(1, 1) = "旧文件名": Cells(1, 2) = "是否删除"
12     strFileName = Dir(strPath & "*.xls*")
13     Do While strFileName <> ""
14         k = k + 1
15         Cells(k, 1) = strPath & strFileName
16         strFileName = Dir
17     Loop
18     Application.DisplayAlerts = True
19 End Sub

```

然后再是否删除那一列标记删除, 将下列代码粘贴到上述代码下面, 然后执行

```

代码块 Sub DeleteFile()
2     Dim r, i As Long
3     r = Range("a1").CurrentRegion '数据装入数组
4     For i = 2 To UBound(r)
5         '标题行不要，从数组第二行开始遍历
6         If r(i, 2) = "删除" Then Kill r(i, 1) 'Kill语句删除指定文件
7     Next
8     MsgBox "完成。"
9 End Sub

```

批量获取指定文件夹下文件名并创建超链接

输入下列代码，选择文件夹

```

代码块
1 Sub GetFiles()
2     Dim strPath As String, strFileName As String, k As Long
3     With Application.FileDialog(msoFileDialogFolderPicker)
4         '用户选择文件夹路径
5         If .Show Then strPath = .SelectedItems(1) Else Exit Sub
6         '如果用户为选择文件夹则退出程序
7     End With
8     If Right(strPath, 1) <> "\" Then strPath = strPath & "\"
9     Application.ScreenUpdating = False '取消屏幕刷新
10    strFileName = Dir(strPath & "*.*)
11    'dir+通配符获取首个文件名
12    '如果一个文件也无，则返回空
13    Columns(1).Clear: Cells(1, 1) = "目录": k = 1 '清除当前工作表A列数据
14    Do While strFileName <> ""
15        k = k + 1 '累加文件个数
16        ActiveSheet.Hyperlinks.Add Cells(k, 1), strPath & strFileName
17        '创建超链接
18        strFileName = Dir
19        '第2次调用Dir函数，未使用任何参数，则同目录下的下一个文件名
20    Loop
21    Application.ScreenUpdating = True
22    MsgBox "一共读取了：" & k - 1 & "个文件名。"
23 End Sub

```

批量给工作簿重命名

在excel空文档输入下列代码，找到对应文件夹，空表就会返回文件名，然后在新文件名那一列改文件名，路径不要动

代码块

```
1 Sub GetFiles()  
2     Dim strPath As String, strFileName As String, k As Long  
3     With Application.FileDialog(msoFileDialogFolderPicker)  
4         If .Show Then strPath = .SelectedItems(1) Else: Exit Sub  
5         '获取用户选择的文件夹的路径, 如果未选取, 则退出程序  
6     End With  
7     If Right(strPath, 1) <> "\" Then strPath = strPath & "\"  
8     Application.ScreenUpdating = False  
9     Range("a:b").Clear: k = 1  
10    '清除A:B列的所有  
11    Cells(1, 1) = "旧文件名": Cells(1, 2) = "新文件名"  
12    strFileName = Dir(strPath & "*.xls*")  
13    Do While strFileName <> ""  
14        k = k + 1  
15        Cells(k, 1) = strPath & strFileName  
16        strFileName = Dir  
17    Loop  
18    Application.DisplayAlerts = True  
19 End Sub
```

改了文件名之后, 输入下列代码, 可以直接在上述代码块下面输入

代码块

```
1 Sub ChangeFileName()  
2     Dim r, i As Long  
3     r = Range("a1").CurrentRegion '数据装入数组  
4     For i = 2 To UBound(r)  
5         '标题行不要, 从数组第二行开始遍历  
6         Name r(i, 1) As r(i, 2) 'Name语句重命名  
7     Next  
8     MsgBox "更名完成。"  
9 End Sub
```

对office程序设置自杀程序

如果有些文件不想被别人破解, 想直接销毁, 就可以用这个程序, 这个文件要保存为启用宏的工作簿, 双击thisworkbook, 关掉, 点保存

代码块

```
1 Private Sub Workbook_Open()  
2     Dim dat As Date
```

```

3      dat = DateSerial(2020, 1, 1)
4      If Date >= dat Then
5          Application.DisplayAlerts = False
6          MsgBox "你是在偷看我的文件吗？" & vbCrLf & "别以为我不知道，我就在你身后看着你！
白衣服，长头发，没有腿的那个。"
7          With ThisWorkbook
8              .Saved = True
9              .ChangeFileAccess xlReadOnly
10             Kill .FullName
11             .Close
12         End With
13     End If
14 End Sub

```

获取多层文件夹下文件名并创建超链接

代码块

```

1  Sub AutoAddLink()
2      Dim strFldPath As String
3      With Application.FileDialog(msoFileDialogFolderPicker)
4          '用户选择指定文件夹
5          .Title = "请选择指定文件夹。"
6          If .Show Then strFldPath = .SelectedItems(1) Else Exit Sub
7          '未选择文件夹则退出程序，否则将地址赋予变量strFldPath
8      End With
9      Application.ScreenUpdating = False
10     '关闭屏幕刷新
11     Range("a:b").ClearContents
12     Range("a1:b1") = Array("文件夹", "文件名")
13     Call SearchFileToHyperlinks(strFldPath)
14     '调取自定义函数SearchFileToHyperlinks
15     Range("a:b").EntireColumn.AutoFit
16     '自动列宽
17     Application.ScreenUpdating = True
18     '重开屏幕刷新
19 End Sub
20 Function SearchFileToHyperlinks(ByVal strFldPath As String) As String
21     Dim objFld As Object
22     Dim objFile As Object
23     Dim objSubFld As Object
24     Dim strFilePath As String
25     Dim lngLastRow As Long
26     Dim intNum As Integer

```

```

27     Set objFld =
CreateObject("Scripting.FileSystemObject").GetFolder(strFldPath)
28     '创建FileSystemObject对象引用
29     For Each objFile In objFld.Files
30     '遍历文件夹内的文件
31         lngLastRow = Cells(Rows.Count, 1).End(xlUp).Row + 1
32         strFilePath = objFile.Path
33         intNum = InStrRev(strFilePath, "\")
34         '使用instrrev函数获取最后文件夹名截至的位置
35         Cells(lngLastRow, 1) = Left(strFilePath, intNum - 1)
36         '文件夹地址
37         Cells(lngLastRow, 2) = Mid(strFilePath, intNum + 1)
38         '文件名
39         ActiveSheet.Hyperlinks.Add Anchor:=Cells(lngLastRow, 2), _
40             Address:=strFilePath, ScreenTip:=strFilePath
41         '添加超链接
42     Next objFile
43     For Each objSubFld In objFld.SubFolders
44     '遍历文件夹内的子文件夹
45         Call SearchFileToHyperlinks(objSubFld.Path)
46     Next objSubFld
47     Set objFld = Nothing
48     Set objFile = Nothing
49     Set objSubFld = Nothing
50 End Function

```

合并多工作簿成总表

字段名要一样，新建空白文档成总表，输入代码，找到对应文件夹，默认汇总全部

代码块

```

1 Sub CollectWorkBookDatas()
2     Dim shtActive As Worksheet, rng As Range, shtData As Worksheet
3     Dim nTitleRow As Long, k As Long, nLastRow As Long
4     Dim i As Long, j As Long, nStartRow As Long
5     Dim aData, aResult, nStarRng As Long
6     Dim strPath As String, strFileName As String
7     Dim strKey As String, nShtCount As Long
8     With Application.FileDialog(msoFileDialogFolderPicker)
9         '取得用户选择的文件夹路径
10         If .Show Then strPath = .SelectedItems(1) Else Exit Sub
11     End With
12     If Right(strPath, 1) <> "\" Then strPath = strPath & "\"
13     strKey = InputBox("请输入需要合并的工作表所包含的关键词：" & vbCrLf & "如未填写关键词，则默认汇总全部表格数据", "提醒")

```

```

14 If StrPtr(strKey) = 0 Then Exit Sub '如果点击了取消或者关闭按钮，则退出程序
15 nTitleRow = Val(InputBox("请输入标题的行数，默认标题行数为1", "提醒", 1))
16 If nTitleRow < 0 Then MsgBox "标题行数不能为负数。", 64, "警告": Exit Sub
17 Set shtActive = ActiveSheet
18 With Application
19     .ScreenUpdating = False
20     .DisplayAlerts = False
21     .AskToUpdateLinks = False
22 End With
23 ReDim aResult(1 To 80000, 1 To 1) '声明结果数组
24 Cells.ClearContents '清空当前表格数据
25 Cells.NumberFormat = "@" '设置单元格为文本格式
26 strFileName = Dir(strPath & "*.xls*") '使用Dir函数遍历excel文件
27 Do While strFileName <> ""
28     If strFileName <> ThisWorkbook.Name Then '避免同名文件重复打开出错
29         With GetObject(strPath & strFileName)
30             '以只读'形式读取文件时，使用getobject会比workbooks.open稍快
31             For Each shtData In .Worksheets '遍历表
32                 If InStr(1, shtData.Name, strKey, vbTextCompare) Then
33                     '如果表中包含关键字则进行汇总(不区分关键词字母大小写)
34                     Set rng = shtData.UsedRange
35                     If rng.Count > 1 Then '判断工作表是否存在数据.....
36                         nShtCount = nShtCount + 1 '汇总工作表的数量
37                         nStartRow = IIf(nShtCount = 1, 1, nTitleRow + 1)
38                     '判断遍历数据源是否应该扣掉标题行
39                     aData = rng.Value '数据区域读入数组arr
40                     If UBound(aData, 2) + 2 > UBound(aResult, 2) Then
41                         '动态调整结果数组brr的最大列数
42                         ReDim Preserve aResult(1 To UBound(aResult), 1
43                         To UBound(aData, 2) + 2)
44                     End If
45                     For i = nStartRow To UBound(aData) '遍历行
46                         k = k + 1
47                         aResult(k, 1) = strFileName '数组第一列放工作簿名
48                         '称
49                         aResult(k, 2) = shtData.Name '数组第二列放工作表名
50                         '称
51                         For j = 1 To UBound(aData, 2) '遍历列
52                             aResult(k, j + 2) = aData(i, j)
53                         Next
54                     If k > UBound(aResult) - 1 Then
55                         '如果数据行数到达结果数组的上限，则将数据导入汇总表，
56                         并清空结果数组
57                     With shtActive
58                         nLastRow = .Cells(Rows.Count,
59                         1).End(xlUp).Row '获取放置来源数据的位置
60                     If nLastRow = 1 Then '判断是否扣除标题行

```

```

54                                     nStarRng = IIf(nTitleRow = 0, 1, 0)
55
56     .Range("a1").Offset(nStarRng).Resize(k, UBound(aResult, 2)) = aResult
57                                     .Range("a1:b1") = Array("来源工作簿名
58 称", "来源工作表名称")
59                                     '前两列放来源工作簿和工作表名称
60                                     Else
61                                     .Range("a1").Offset(nLastRow).Resize(k, UBound(aResult, 2)) = aResult
62                                     '放结果数组的数据
63                                     End If
64                                     End With
65                                     k = 0
66                                     ReDim aResult(1 To UBound(aResult), 1 To
67 UBound(aResult, 2))
68                                     '重新设置结果数组
69                                     End If
70                                     Next
71                                     End If
72                                     End If
73                                     Next
74                                     .Close False '关闭工作簿
75                                     End With
76                                     End If
77                                     strFileName = Dir '下一个excel文件
78                                     Loop
79                                     If k > 0 Then
80                                         shtActive.Select '激活汇总表
81                                         nLastRow = Cells(Rows.Count, 1).End(xlUp).Row '放置数据的位置
82                                         If nLastRow = 1 Then '如果汇总表数据为空, 说明需要汇总的数据没有超过结果数组的
83 上限
84                                             nStarRng = IIf(nTitleRow = 0, 1, 0)
85                                             Range("a1").Offset(nStarRng).Resize(k, UBound(aResult, 2)) =
86 aResult
87                                             Range("a1:b1") = Array("来源工作簿名称", "来源工作表名称")
88                                             Else
89                                             Range("a1").Offset(nLastRow).Resize(k, UBound(aResult, 2)) =
90 aResult
91                                             End If
92                                         End If
93                                         With Application
94                                             .ScreenUpdating = True
95                                             .DisplayAlerts = True
96                                             .AskToUpdateLinks = True
97                                         End With
98                                         MsgBox "一共汇总完成。" & nShtCount & "个工作表", , "孙兴华"
99                                     End Sub

```


将word表格批量写入excel

打开excel空白文件，输入代码，找到word文档

代码块

```
1 Sub GetWordTable()  
2     Dim WdApp As Object  
3     Dim objTable As Object  
4     Dim objDoc As Object  
5     Dim strPath As String  
6     Dim shtEach As Worksheet  
7     Dim shtSelect As Worksheet  
8     Dim i As Long  
9     Dim j As Long  
10    Dim x As Long  
11    Dim y As Long  
12    Dim k As Long  
13    Dim brr As Variant  
14    Set WdApp = CreateObject("Word.Application")  
15    With Application.FileDialog(msoFileDialogFilePicker)  
16        .Filters.Add "Word文件", "*.doc*", 1  
17        '只显示word文件  
18        .AllowMultiSelect = False  
19        '禁止多选文件  
20        If .Show Then strPath = .SelectedItems(1) Else Exit Sub  
21    End With  
22    Application.ScreenUpdating = False  
23    Application.DisplayAlerts = False  
24    Set shtSelect = ActiveSheet  
25    '当前表赋值变量shtSelect，方便代码运行完成后叶落归根回到开始的地方  
26    For Each shtEach In Worksheets  
27        '删除当前工作表以外的所有工作表  
28        If shtEach.Name <> shtSelect.Name Then shtEach.Delete  
29    Next  
30    shtSelect.Name = "孙兴华"  
31    '这句代码不是无聊，作用在于.....你猜.....  
32    '.....其实是避免下面的程序工作表名称重复  
33    Set objDoc = WdApp.documents.Open(strPath)  
34    '后台打开用户选定的word文档  
35    For Each objTable In objDoc.tables  
36        '遍历文档中的每个表格  
37        k = k + 1  
38        Worksheets.Add after:=Worksheets(Worksheets.Count)  
39        '新建工作表  
40        ActiveSheet.Name = k & "表"
```

```

41         x = objTable.Rows.Count
42         'table的行数
43         y = objTable.Columns.Count
44         'table的列数
45         ReDim brr(1 To x, 1 To y)
46         '以下遍历行列，数据写入数组brr
47         For i = 1 To x
48             For j = 1 To y
49                 brr(i, j) = "" & Application.Clean(objTable.cell(i,
j).Range.Text)
50                 'Clean函数清除制表符等
51                 '半角单引号将数据统一转换为文本格式，避免身份证等数值变形
52             Next
53         Next
54         With [a1].Resize(x, y)
55             .Value = brr
56             '数据写入Excel工作表
57             .Borders.LineStyle = 1
58             '添加边框线
59         End With
60     Next
61     shtSelect.Select
62     objDoc.Close: WdApp.Quit
63     Application.ScreenUpdating = True
64     Application.DisplayAlerts = True
65     Set objDoc = Nothing
66     Set WdApp = Nothing
67     MsgBox "共获取：" & k & "张表格的数据。"
68 End Sub

```

取消复杂的合并单元格

取消后可直接填充好，选择区域

代码块

```

1 Sub UnMergeRange2() '取消合并单元格
2 Dim MaxRow As Integer '
3 Dim Rng As Range
4 Dim x%, y%, m%, n%, i%
5 Dim Rng2 As Range
6 On Error Resume Next
7 Set Rng = Application.InputBox("请选择需要取消合并单元格的区域：" & _
8     "区域选择", , , , , 8)
9
10 For x = 1 To Rng.Rows.Count

```

```

11         For y = 1 To Rng.Columns.Count
12             Set Rng2 = Rng.Cells(x, y)
13             i = Rng2.MergeArea.Count
14             If i > 1 Then
15                 m = Rng2.MergeArea.Rows.Count
16                 n = Rng2.MergeArea.Columns.Count
17                 Rng2.UnMerge '取消合并单元格
18                 Rng2.Resize(m, n).Value = Rng2.Value
19             End If
20         Next
21     Next
22
23 End Sub

```

批量将图片插入到单元格批注中

照片命名跟excel里面各单元格的命名要匹配,要找到照片文件夹,要找到插入区域

代码块

```

1 Sub AddCommentPic()
2     Dim arr, i&, k&, n&, b As Boolean
3     Dim strPicName$, strPicPath$, strFdPath$
4     Dim rngData As Range, rngEach As Range
5     'On Error Resume Next
6     '用户选择图片所在的文件夹
7     With Application.FileDialog(msoFileDialogFolderPicker)
8         If .Show Then strFdPath = .SelectedItems(1) Else: Exit Sub
9     End With
10    If Right(strFdPath, 1) <> "\" Then strFdPath = strFdPath & "\"
11    Set rngData = Application.InputBox("请选择需要插入图片到批注中的单元格区域",
Type:=8)
12    '用户选择需要插入图片到批注中的单元格或区域
13    If rngData.Count = 0 Then Exit Sub
14    Set rngData = Intersect(rngData.Parent.UsedRange, rngData)
15    'intersect语句避免用户选择整列单元格,造成无谓运算的情况
16    If rngData Is Nothing Then MsgBox "选择单元格不能全为空。": Exit Sub
17    arr = Array(".jpg", ".jpeg", ".bmp", ".png", ".gif")
18    '用数组变量记录五种文件格式
19    Application.ScreenUpdating = False
20    For Each rngEach In rngData
21        '遍历选择区域的每一个单元格
22        If Not rngEach.Comment Is Nothing Then rngEach.Comment.Delete '删除旧的
批注

```

```

23         strPicName = rngEach.Text '图片名称
24         If Len(strPicName) Then '如果单元格存在值
25             strPicPath = strFdPath & strPicName '图片路径
26             b = False 'pd变量标记是否找到相关图片
27             For i = 0 To UBound(arr)
28                 '由于不确定用户的图片格式，因此遍历图片格式
29                 If Len(Dir(strPicPath & arr(i))) Then
30                     '如果存在相关文件
31                     rngEach.AddComment '增加批注
32                     With rngEach.Comment
33                         .Visible = True '批注可见
34                         .Text Text:=""
35                         .Shape.Select True '选中批注图形
36                         Selection.ShapeRange.Fill.UserPicture strPicPath &
arr(i)
37                         '插入图片到批注中
38                         .Shape.Height = 150 '图形的高度，可以根据需要自己调整
39                         .Shape.Width = 150 '图形的宽度，可以根据需要自己调整
40                         .Visible = False '取消显示
41                     End With
42                     b = True '标记找到结果
43                     n = n + 1 '累加找到结果的个数
44                     Exit For '找到结果后就可以退出文件格式循环
45                 End If
46             Next
47             If b = False Then k = k + 1 '如果没找到图片累加个数
48         End If
49     Next
50     MsgBox "共处理成功" & n & "个图片，另有" & k & "个非空单元格未找到对应的图片。"
51     Application.ScreenUpdating = True
52 End Sub

```

批量将图片插入到表格中

代码块

```

1 Sub InsertPic()
2     Dim arr, i&, k&, n&, b As Boolean
3     Dim strPicName$, strPicPath$, strFdPath$, shp As Shape
4     Dim rngData As Range, rngEach As Range, rngWhere As Range, strWhere As
String
5     'On Error Resume Next
6     '用户选择图片所在的文件夹
7     With Application.FileDialog(msoFileDialogFolderPicker)
8         If .Show Then strFdPath = .SelectedItems(1) Else: Exit Sub
9     End With

```

```

10 If Right(strFdPath, 1) <> "\" Then strFdPath = strFdPath & "\"
11 Set rngData = Application.InputBox("请选择图片名称所在的单元格区域", Type:=8)
12 '用户选择需要插入图片的名称所在单元格范围
13 Set rngData = Intersect(rngData.Parent.UsedRange, rngData)
14 'intersect语句避免用户选择整列单元格, 造成无谓运算的情况
15 If rngData Is Nothing Then MsgBox "选择的单元格范围不存在数据! ": Exit Sub
16 strWhere = InputBox("请输入图片偏移的位置, 例如上1、下1、左1、右1", , "右1")
17 '用户输入图片相对单元格的偏移位置。
18 If Len(strWhere) = 0 Then Exit Sub
19 x = Left(strWhere, 1)
20 '偏移的方向
21 If InStr("上下左右", x) = 0 Then MsgBox "你未输入偏移方位。": Exit Sub
22 y = Val(Mid(strWhere, 2))
23 '偏移的值
24 Select Case x
25     Case "上"
26         Set rngWhere = rngData.Offset(-y, 0)
27     Case "下"
28         Set rngWhere = rngData.Offset(y, 0)
29     Case "左"
30         Set rngWhere = rngData.Offset(0, -y)
31     Case "右"
32         Set rngWhere = rngData.Offset(0, y)
33 End Select
34 Application.ScreenUpdating = False
35 rngData.Parent.Parent.Activate '用户选定的激活工作簿
36 rngData.Parent.Select
37 For Each shp In ActiveSheet.Shapes
38     '如果旧图片存放在目标图片存放范围则删除
39     If Not Intersect(rngWhere, shp.TopLeftCell) Is Nothing Then shp.Delete
40 Next
41 x = rngWhere.Row - rngData.Row
42 y = rngWhere.Column - rngData.Column
43 '偏移的坐标
44 arr = Array(".jpg", ".jpeg", ".bmp", ".png", ".gif")
45 '用数组变量记录五种文件格式
46 For Each rngEach In rngData
47     '遍历选择区域的每一个单元格
48     strPicName = rngEach.Text
49     '图片名称
50     If Len(strPicName) Then
51         '如果单元格存在值
52         strPicPath = strFdPath & strPicName
53         '图片路径
54         b = False
55         '变量标记是否找到相关图片
56         For i = 0 To UBound(arr)

```

```

57         '由于不确定用户的图片格式，因此遍历图片格式
58         If Len(Dir(strPicPath & arr(i))) Then
59             '如果存在相关文件
60             Set shp = ActiveSheet.Shapes.AddPicture( _
61                 strPicPath & arr(i), False, True, _
62                 rngEach.Offset(x, y).Left + 5, _
63                 rngEach.Offset(x, y).Top + 5, _
64                 20, 20)
65             shp.Select
66             With Selection
67                 .ShapeRange.LockAspectRatio = msoFalse
68                 '撤销锁定图片纵横比
69                 .Height = rngEach.Offset(x, y).Height - 10 '图片高度
70                 .Width = rngEach.Offset(x, y).Width - 10 '图片宽度
71             End With
72             b = True '标记找到结果
73             n = n + 1 '累加找到结果的个数
74             Range("a1").Select: Exit For '找到结果后就可以退出文件格式循环
75         End If
76     Next
77     If b = False Then k = k + 1 '如果没找到图片累加个数
78 End If
79 Next
80 Application.ScreenUpdating = True
81 MsgBox "共处理成功" & n & "个图片，另有" & k & "个非空单元格未找到对应的图片。"
82 End Sub

```

单元格被修改后被批注记录

代码块

```

1  '在所有过程之前用Dim语句定义的变量r1是模块级变量，应模块中所有的过程都可以使用它
2  Dim r1 '定义一个模块级变量，用户保存单元格的数据
3  '第一个事件过程，用于记录被更改前单元格中保存的数据
4  Private Sub Worksheet_SelectionChange(ByVal Target As Range)
5      If Target.Cells.Count <> 1 Then Exit Sub '选中多个单元格时退出程序
6      If Target.Formula = "" Then '根据选中单元格中保存的数据，确定给变量r1赋什么值
7          r1 = "空"
8      Else
9          r1 = Target.Text
10     End If
11 End Sub
12 '第二个事件过程，用于批注记录单元格修改前后的信息
13 Private Sub Worksheet_Change(ByVal Target As Range)
14     If Target.Cells.Count <> 1 Then Exit Sub
15     '定义变量保存单元格修改后的内容

```

```

16 Dim r2
17 '判断单元格是否被修改为空单元格
18 If Target.Formula = "" Then
19     r2 = "空"
20 Else
21     r2 = Target.Formula
22 End If
23 '如果单元格修改前后的内容一样则退出程序
24 If r1 = r2 Then Exit Sub
25 '定义一个批注变量
26 Dim r3
27 '定义一个变量保存批注内容
28 Dim r4
29 '将被修改单元格的批注赋给变量r3
30 Set r3 = Target.Comment
31 '如果单元格中没有批注则新建批注
32 If r3 Is Nothing Then Target.AddComment
33 '将批注的内容保存到变量r4中
34 r4 = Target.Comment.Text
35 '重新修改批注的内容=原批注内容+当前日期和时间+原内容+修改后的新内容
36 Target.Comment.Text Text:=r4 & Chr(10) & Format(Now(), "yyyy-mm-dd hh:mm") &
    "原内容:" & r1 & "修改为:" & r2
37 '根据批注内容自动调整批注大小
38 Target.Comment.Shape.TextFrame.AutoSize = True
39 End Sub

```

excel自动保存

下列代码右键新建模块输入

代码块

```

1 Sub otime()
2     '10秒后自动运行WbSave过程
3     Application.OnTime Now() + TimeValue("00:00:10"), "WbSave"
4 End Sub
5 Sub WbSave()
6     ThisWorkbook.Save '保存本工作簿
7     Call otime '再次运行otime过程
8 End Sub

```

下列代码双击thisworkbook输入

代码块

```
1 Private Sub Workbook_Open()  
2 Call otime  
3 End Sub
```

自动化报表（录制宏）

操作步骤：如果有数据透视表的操作，需要先建一个数据透视表，因为怕后续excel找不到这个表，然后开发工具-录制宏（如果是在一个表格里面的重复操作就先点相对引用之后再录制），直接给宏命名之后点确定，等操作完成之后再停止录制，等下次要做相同操作时再执行宏就好。但是需要注意的是，这可能没办法很灵活，针对不同的数据量可能需要手动修改代码

if语句与for循环

While循环与嵌套

工作表

aplication-excel应用程序，workbook工作簿，worksheet工作表，range单元格

对象

最好分步写代码，因为好修改且好做窗体控件

VBA函数

存放每个元素得单元格

选择

赋值

清除赋值

单元格字体等处理，合并单元格

激活单元格

复制粘贴和删除

数据类型与日期函数